###############################################################################
@mainSQL[][tObjectInfo;oModule;hXSLTParam]

^rem{ свойства страницы, которая соответствует текущему URI }
$tObjectInfo[^MAIN:core.getObject[]]

^if(def $tObjectInfo.id){

  ^rem{ ### ВЫЗОВ МОДУЛЯ СТРАНИЦЫ ### }
  $oModule[^factory::create[$tObjectInfo.module][
		$.id[$tObjectInfo.id]
		$.pSQL[$MAIN:pSQL]
		$.core[$MAIN:core]
	]]
  $xContent[^oModule.xContent[]]

	^rem{ параметры шаблона  }
	$hXSLTParam[^oModule.XMLparam[]]

  ^rem{ секундомер на время XSLT-преобразования }
	$xslt_start_time($status:rusage.tv_sec+$status:rusage.tv_usec/1000000)

	^rem{ отладочный вывод или чистовой с применением шаблона }
	^switch[$form:debug]{
		^case[xml]{
			^rem{ выключено кеширование. Вывод XML документа, полученного из модуля. }
			$response:expires[^date::now(-1)]
			$response:Pragma[no-cache]
			$response:Cache-Control[private,no-cache,no-store]
			^use[ixColorer.p]
			$result[^taint[html][^xContent.string[$.method[xml]]]]
			^rem{ подсветка синтаксиса xml }
			$pColor[^ixColorer::create[$result]]
			$result[^pColor.getString[]]
			$result[^result.match[[\n]+[\s]*[\n]+][g]{<br>}]
			$result[<html><head><link rel="stylesheet" type="text/css" href="/admin/css/highlighting.css" /></head>
							<body>^result.match[[\t]+][g]{&nbsp^;&nbsp^;&nbsp^;&nbsp^;}</body></html>]
		}
		^case[html]{
			$html[^xContent.transform[/admin/templates/${tObjectInfo.template}.xsl][$hXSLTParam]]
			$result[<pre>^taint[html][^html.string[$.method[html]]]</pre>]
			$result[^result.match[[\n]+[\s]*[\n]+][g]{<br>}]
			$result[^result.match[[\t]+][g]{&nbsp^;&nbsp^;&nbsp^;&nbsp^;}]
		}
		^case[plain]{
			^rem{ вывод в html + выключено кеширование + выводится статистика времени работы }
			$response:expires[^date::now(-1)]
			$response:Pragma[no-cache]
			$response:Cache-Control[private,no-cache,no-store]
			$html[^xContent.transform[/admin/templates/${tObjectInfo.template}.xsl][$hXSLTParam]]
			$result[^html.string[$.method[html]]]
		}
		^case[DEFAULT]{
			$html[^xContent.transform[/admin/templates/${tObjectInfo.template}.xsl][$hXSLTParam]]
			$result[^html.string[$.method[html]]]
		}
	}
	$xslt_exec_time($status:rusage.tv_sec + $status:rusage.tv_usec/1000000 - $xslt_start_time)
	$timer2[XSLT: ^xslt_exec_time.format[%.3f] с, ]
	$result[^result.match[time2time2][]{^taint[xml][$timer2]}]

}{
  ^rem{ если объект не найден }
  $result[<h3>Объект не найден (<font color='red'>404</font>) $request:uri</h3>]
}
###############################################################################